Rules for integrands of the form $u\left(a+b x+c x^2+d x^3\right)^p$

1. $\int u\left(a+b x+d x^3\right)^p\,dx$

   1. $\int\left(a+b x+d x^3\right)^p\,dx$

      1. $\int\left(a+b x+d x^3\right)^p\,dx$ when $p\in\mathbb{Z}$

         1: $\int\left(a+b x+d x^3\right)^p\,dx$ when $p\in\mathbb{Z}\land 4 b^3+27 a^2 d=0$

- **Derivation: Algebraic expansion**
- **Basis: If** $4 b^3+27 a^2 d=0$**, then** $a+b x+d x^3=\frac{1}{3^3 a^2}(3 a-b x)(3 a+2 b x)^2$
- **Rule: If** $p\in\mathbb{Z}\land 4 b^3+27 a^2 d=0$**, then**

$$\int\left(a+b x+d x^3\right)^p\,dx\to\frac{1}{3^{3 p} a^{2 p}}\int(3 a-b x)^p(3 a+2 b x)^{2 p}\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^(3*p)*a^(2*p))*Int[(3*a-b*x)^p*(3*a+2*b*x)^(2*p),x] /;
FreeQ[{a,b,d},x] && IntegerQ[p] && EqQ[4*b^3+27*a^2*d,0]
```

2. $\int\left(a+b x+d x^3\right)^p\,dx$ when $p\in\mathbb{Z}\land 4 b^3+27 a^2 d\neq 0$

   1: $\int\left(a+b x+d x^3\right)^p\,dx$ when $p\in\mathbb{Z}^+\land 4 b^3+27 a^2 d\neq 0$

- **Derivation: Algebraic expansion**
- **Rule: If** $p\in\mathbb{Z}^+\land 4 b^3+27 a^2 d\neq 0$**,**

$$\int\left(a+b x+d x^3\right)^p\,dx\to\int\text{ExpandToSum}\left[\left(a+b x+d x^3\right)^p,x\right]dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  Int[ExpandToSum[(a+b*x+d*x^3)^p,x],x] /;
FreeQ[{a,b,d},x] && IGtQ[p,0] && NeQ[4*b^3+27*a^2*d,0]
```

2. $\int (a + b x + d x^3)^p \, dx$ when $p \in \mathbb{Z}^- \land 4 b^3 + 27 a^2 d \neq 0$

1: $\int (a + b x + d x^3)^p \, dx$ when $p \in \mathbb{Z}^- \land 4 b^3 + 27 a^2 d \neq 0 \land \text{Factor}[a + b x + d x^3] == v w \cdots$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^- \land 4 b^3 + 27 a^2 d \neq 0 \land \text{Factor}[a + b x + d x^3] == v w \cdots$**, then**

$$\int (a + b x + d x^3)^p \, dx \longrightarrow \int v^p w^p \cdots \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=Factor[a+b*x+d*x^3]},
  FreeFactors[u,x]^p*Int[DistributeDegree[NonfreeFactors[u,x],p],x] /;
 ProductQ[NonfreeFactors[u,x]]] /;
FreeQ[{a,b,d},x] && ILtQ[p,0] && NeQ[4*b^3+27*a^2*d,0]
```

2: $\int (a + b x + d x^3)^p \, dx$ when $p \in \mathbb{Z}^- \wedge 4 b^3 + 27 a^2 d \neq 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $r = \left(-27 a d^2 + 3 \sqrt{3} \, d \sqrt{4 b^3 d + 27 a^2 d^2}\right)^{1/3}$**, then**

$$a + b x + d x^3 == \frac{1}{3^3 d^2} \left(\frac{6 b d - 2^{1/3} r^2}{2^{2/3} r} + 3 d x\right) \left(\frac{6 \left(1 + i \sqrt{3}\right) b d - 2^{1/3} \left(1 - i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} - 3 d x\right) \left(\frac{6 \left(1 - i \sqrt{3}\right) b d - 2^{1/3} \left(1 + i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} - 3 d x\right)$$

- **Rule: If** $p \in \mathbb{Z}^- \wedge 4 b^3 + 27 a^2 d \neq 0$**, let** $r = \left(-27 a d^2 + 3 \sqrt{3} \, d \sqrt{4 b^3 d + 27 a^2 d^2}\right)^{1/3}$**, then**

$$\int (a + b x + d x^3)^p \, dx \longrightarrow$$

$$\frac{1}{3^{3 p} d^{2 p}} \int \left(\frac{6 b d - 2^{1/3} r^2}{2^{2/3} r} + 3 d x\right)^p \left(\frac{6 \left(1 + i \sqrt{3}\right) b d - 2^{1/3} \left(1 - i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} - 3 d x\right)^p \left(\frac{6 \left(1 - i \sqrt{3}\right) b d - 2^{1/3} \left(1 + i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} - 3 d x\right)^p dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-27*a*d^2+3*Sqrt[3]*d*Sqrt[4*b^3*d+27*a^2*d^2],3]},
  1/(3^(3*p)*d^(2*p))*
    Int[((6*b*d-2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      ((6*(1+I*Sqrt[3])*b*d-2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p*
      ((6*(1-I*Sqrt[3])*b*d-2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p,x]] /;
FreeQ[{a,b,d},x] && ILtQ[p,0] && NeQ[4*b^3+27*a^2*d,0]
```

2. $\int (a + b x + d x^3)^p \, dx$ when $p \notin \mathbb{Z}$

1: $\int (a + b x + d x^3)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 b^3 + 27 a^2 d == 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If $4 b^3 + 27 a^2 d == 0$, then $\partial_x \frac{(a+b x+d x^3)^p}{(3 a-b x)^p (3 a+2 b x)^{2 p}} == 0$**
- **Rule: If $p \notin \mathbb{Z} \wedge 4 b^3 + 27 a^2 d == 0$, then**

$$\int (a + b x + d x^3)^p \, dx \longrightarrow \frac{(a + b x + d x^3)^p}{(3 a - b x)^p (3 a + 2 b x)^{2 p}} \int (3 a - b x)^p (3 a + 2 b x)^{2 p} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  (a+b*x+d*x^3)^p/((3*a-b*x)^p*(3*a+2*b*x)^(2*p))*Int[(3*a-b*x)^p*(3*a+2*b*x)^(2*p),x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]] && EqQ[4*b^3+27*a^2*d,0]
```

2. $\int (a + b x + d x^3)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 b^3 + 27 a^2 d \neq 0$

1: $\int (a + b x + d x^3)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 b^3 + 27 a^2 d \neq 0 \wedge \text{Factor}[a + b x + d x^3] == v w \cdots$

- **Derivation: Piecewise constant extraction**
- **Basis: If $\text{Factor}[a + b x + d x^3] == v w \cdots$, then $\partial_x \frac{(a+b x+d x^3)^p}{v^p w^p \cdots} == 0$**
- **Rule: If $p \notin \mathbb{Z} \wedge 4 b^3 + 27 a^2 d \neq 0 \wedge \text{Factor}[a + b x + d x^3] == v w \cdots$, then**

$$\int (a + b x + d x^3)^p \, dx \longrightarrow \frac{(a + b x + d x^3)^p}{v^p w^p \cdots} \int v^p w^p \cdots \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=NonfreeFactors[Factor[a+b*x+d*x^3],x]},
  (a+b*x+d*x^3)^p/DistributeDegree[u,p]*Int[DistributeDegree[u,p],x] /;
 ProductQ[u]] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]] && NeQ[4*b^3+27*a^2*d,0]
```

2: $\int (a+b\,x+d\,x^3)^p\,dx$ when $p\notin\mathbb{Z}\ \land\ 4\,b^3+27\,a^2\,d\neq 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $r=\left(-27\,a\,d^2+3\,\sqrt{3}\,d\,\sqrt{4\,b^3\,d+27\,a^2\,d^2}\right)^{1/3}$**, then**

$$\partial_x \frac{(a+b\,x+d\,x^3)^p}{\left(\left(\frac{6\,b\,d-2^{1/3}\,r^2}{2^{2/3}\,r}+3\,d\,x\right)^p\left(\frac{6\,\left(1+i\,\sqrt{3}\right)\,b\,d-2^{1/3}\,\left(1-i\,\sqrt{3}\right)\,r^2}{2\times 2^{2/3}\,r}-3\,d\,x\right)^p\left(\frac{6\,\left(1-i\,\sqrt{3}\right)\,b\,d-2^{1/3}\,\left(1+i\,\sqrt{3}\right)\,r^2}{2\times 2^{2/3}\,r}-3\,d\,x\right)^p\right)}==0$$

- **Rule: If** $p\notin\mathbb{Z}\ \land\ 4\,b^3+27\,a^2\,d\neq 0$**, let** $r=\left(-27\,a\,d^2+3\,\sqrt{3}\,d\,\sqrt{4\,b^3\,d+27\,a^2\,d^2}\right)^{1/3}$**, then**

$$\int (a+b\,x+d\,x^3)^p\,dx\ \longrightarrow$$

$$\frac{(a+b\,x+d\,x^3)^p}{\left(\left(\frac{6\,b\,d-2^{1/3}\,r^2}{2^{2/3}\,r}+3\,d\,x\right)^p\left(\frac{6\,\left(1+i\,\sqrt{3}\right)\,b\,d-2^{1/3}\,\left(1-i\,\sqrt{3}\right)\,r^2}{2\times 2^{2/3}\,r}-3\,d\,x\right)^p\left(\frac{6\,\left(1-i\,\sqrt{3}\right)\,b\,d-2^{1/3}\,\left(1+i\,\sqrt{3}\right)\,r^2}{2\times 2^{2/3}\,r}-3\,d\,x\right)^p\right)}\cdot$$

$$\int\left(\frac{6\,b\,d-2^{1/3}\,r^2}{2^{2/3}\,r}+3\,d\,x\right)^p\left(\frac{6\,\left(1+i\,\sqrt{3}\right)\,b\,d-2^{1/3}\,\left(1-i\,\sqrt{3}\right)\,r^2}{2\times 2^{2/3}\,r}-3\,d\,x\right)^p\left(\frac{6\,\left(1-i\,\sqrt{3}\right)\,b\,d-2^{1/3}\,\left(1+i\,\sqrt{3}\right)\,r^2}{2\times 2^{2/3}\,r}-3\,d\,x\right)^p\,dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-27*a*d^2+3*Sqrt[3]*d*Sqrt[4*b^3*d+27*a^2*d^2],3]},
  (a+b*x+d*x^3)^p/(((6*b*d-2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      ((6*(1+I*Sqrt[3])*b*d-2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p*
      ((6*(1-I*Sqrt[3])*b*d-2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p)*
    Int[((6*b*d-2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      ((6*(1+I*Sqrt[3])*b*d-2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p*
      ((6*(1-I*Sqrt[3])*b*d-2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p,x]] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]] && NeQ[4*b^3+27*a^2*d,0]
```

2. $\int (e + f x)^m \left(a + b x + d x^3\right)^p dx$

1. $\int (e + f x)^m \left(a + b x + d x^3\right)^p dx$ when $p \in \mathbb{Z}$

1: $\int (e + f x)^m \left(a + b x + d x^3\right)^p dx$ when $p \in \mathbb{Z} \land 4 b^3 + 27 a^2 d == 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $4 b^3 + 27 a^2 d == 0$**, then** $a + b x + d x^3 == \frac{1}{3^3 a^2} (3 a - b x) (3 a + 2 b x)^2$
- **Rule: If** $p \in \mathbb{Z} \land 4 b^3 + 27 a^2 d == 0$**, then**

$$\int (e + f x)^m \left(a + b x + d x^3\right)^p dx \longrightarrow \frac{1}{3^{3 p} a^{2 p}} \int (e + f x)^m (3 a - b x)^p (3 a + 2 b x)^{2 p} dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^(3*p)*a^(2*p))*Int[(e+f*x)^m*(3*a-b*x)^p*(3*a+2*b*x)^(2*p),x] /;
FreeQ[{a,b,d,e,f,m},x] && IntegerQ[p] && EqQ[4*b^3+27*a^2*d,0]
```

2. $\int (e + f x)^m \left(a + b x + d x^3\right)^p dx$ when $p \in \mathbb{Z} \land 4 b^3 + 27 a^2 d \neq 0$

1: $\int (e + f x)^m \left(a + b x + d x^3\right)^p dx$ when $p \in \mathbb{Z}^+ \land 4 b^3 + 27 a^2 d \neq 0$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^+ \land 4 b^3 + 27 a^2 d \neq 0$**,**

$$\int (e + f x)^m \left(a + b x + d x^3\right)^p dx \longrightarrow \int \text{ExpandIntegrand}\left[(e + f x)^m \left(a + b x + d x^3\right)^p, x\right] dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(e+f*x)^m*(a+b*x+d*x^3)^p,x],x] /;
FreeQ[{a,b,d,e,f,m},x] && IGtQ[p,0] && NeQ[4*b^3+27*a^2*d,0]
```

2. $\int (e+f x)^m \left(a+b x+d x^3\right)^p dx$ when $p \in \mathbb{Z}^- \land 4 b^3+27 a^2 d \neq 0$

1: $\int (e+f x)^m \left(a+b x+d x^3\right)^p dx$ when $p \in \mathbb{Z}^- \land 4 b^3+27 a^2 d \neq 0 \land \text{Factor}\left[a+b x+d x^3\right] == v\, w \cdots$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^- \land 4 b^3+27 a^2 d \neq 0 \land \text{Factor}\left[a+b x+d x^3\right] == v\, w \cdots$**, then**

$$\int (e+f x)^m \left(a+b x+d x^3\right)^p dx \longrightarrow \int (e+f x)^m v^p w^p \cdots dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=Factor[a+b*x+d*x^3]},
  FreeFactors[u,x]^p*Int[(e+f*x)^m*DistributeDegree[NonfreeFactors[u,x],p],x] /;
 ProductQ[NonfreeFactors[u,x]]] /;
FreeQ[{a,b,d,e,f,m},x] && ILtQ[p,0] && NeQ[4*b^3+27*a^2*d,0]
```

2: $\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx$ when $p \in \mathbb{Z}^- \land 4 b^3+27 a^2 d \neq 0$

- **Derivation: Algebraic expansion**

- **Basis: If** $r = \left(-27 a d^2+3 \sqrt{3} \, d \sqrt{4 b^3 d+27 a^2 d^2}\right)^{1/3}$**, then**

$$a+b x+d x^3 == \frac{1}{3^3 d^2} \left(\frac{6 b d-2^{1/3} r^2}{2^{2/3} r}+3 d x\right) \left(\frac{6 \left(1+i \sqrt{3}\right) b d-2^{1/3} \left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right) \left(\frac{6 \left(1-i \sqrt{3}\right) b d-2^{1/3} \left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)$$

- **Rule: If** $p \in \mathbb{Z}^- \land 4 b^3+27 a^2 d \neq 0$**, let** $r = \left(-27 a d^2+3 \sqrt{3} \, d \sqrt{4 b^3 d+27 a^2 d^2}\right)^{1/3}$**, then**

$$\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx \longrightarrow$$

$$\frac{1}{3^{3 p} d^{2 p}} \int (e+f x)^m \left(\frac{6 b d-2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(\frac{6 \left(1+i \sqrt{3}\right) b d-2^{1/3} \left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)^p \left(\frac{6 \left(1-i \sqrt{3}\right) b d-2^{1/3} \left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)^p dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-27*a*d^2+3*Sqrt[3]*d*Sqrt[4*b^3*d+27*a^2*d^2],3]},
  1/(3^(3*p)*d^(2*p))*
    Int[(e+f*x)^m*((6*b*d-2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      ((6*(1+I*Sqrt[3])*b*d-2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p*
      ((6*(1-I*Sqrt[3])*b*d-2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p,x]] /;
FreeQ[{a,b,d,e,f,m},x] && ILtQ[p,0] && NeQ[4*b^3+27*a^2*d,0]
```

2. $\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z}$

1: $\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 b^3+27 a^2 d == 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If $4 b^3+27 a^2 d == 0$, then $\partial_x \frac{(a+b x+d x^3)^p}{(3 a-b x)^p (3 a+2 b x)^{2 p}} == 0$**
- **Rule: If $p \notin \mathbb{Z} \wedge 4 b^3+27 a^2 d == 0$, then**

$$\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx \longrightarrow \frac{\left(a+b x+d x^3\right)^p}{(3 a-b x)^p (3 a+2 b x)^{2 p}} \int (e+f x)^m (3 a-b x)^p (3 a+2 b x)^{2 p} \, dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  (a+b*x+d*x^3)^p/((3*a-b*x)^p*(3*a+2*b*x)^(2*p))*Int[(e+f*x)^m*(3*a-b*x)^p*(3*a+2*b*x)^(2*p),x] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[p]] && EqQ[4*b^3+27*a^2*d,0]
```

2. $\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 b^3+27 a^2 d \neq 0$

1: $\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 b^3+27 a^2 d \neq 0 \wedge \text{Factor}\left[a+b x+d x^3\right] == v\, w \cdots$

- **Derivation: Piecewise constant extraction**
- **Basis: If $\text{Factor}\left[a+b x+d x^3\right] == v\, w \cdots$, then $\partial_x \frac{(a+b x+d x^3)^p}{v^p w^p \cdots} == 0$**
- **Rule: If $p \notin \mathbb{Z} \wedge 4 b^3+27 a^2 d \neq 0 \wedge \text{Factor}\left[a+b x+d x^3\right] == v\, w \cdots$, then**

$$\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx \longrightarrow \frac{\left(a+b x+d x^3\right)^p}{v^p w^p \cdots} \int (e+f x)^m v^p w^p \cdots \, dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=NonfreeFactors[Factor[a+b*x+d*x^3],x]},
  (a+b*x+d*x^3)^p/DistributeDegree[u,p]*Int[(e+f*x)^m*DistributeDegree[u,p],x] /;
 ProductQ[u]] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[p]] && NeQ[4*b^3+27*a^2*d,0]
```

2: $\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \land 4 b^3+27 a^2 d \neq 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $r=\left(-27 a d^2+3 \sqrt{3} d \sqrt{4 b^3 d+27 a^2 d^2}\right)^{1/3}$**, then**

$$\partial_x \left(a+b x+d x^3\right)^p \Big/ \left(\left(\frac{6 b d-2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(\frac{6 \left(1+i \sqrt{3}\right) b d-2^{1/3} \left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)^p \left(\frac{6 \left(1-i \sqrt{3}\right) b d-2^{1/3} \left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)^p\right) = 0$$

- **Rule: If** $p \notin \mathbb{Z} \land 4 b^3+27 a^2 d \neq 0$**, let** $r=\left(-27 a d^2+3 \sqrt{3} d \sqrt{4 b^3 d+27 a^2 d^2}\right)^{1/3}$**, then**

$$\int (e+f x)^m \left(a+b x+d x^3\right)^p \, dx \rightarrow$$

$$\left(a+b x+d x^3\right)^p \Big/ \left(\left(\frac{6 b d-2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(\frac{6 \left(1+i \sqrt{3}\right) b d-2^{1/3} \left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)^p \left(\frac{6 \left(1-i \sqrt{3}\right) b d-2^{1/3} \left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)^p\right) \cdot$$

$$\int (e+f x)^m \left(\frac{6 b d-2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(\frac{6 \left(1+i \sqrt{3}\right) b d-2^{1/3} \left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)^p \left(\frac{6 \left(1-i \sqrt{3}\right) b d-2^{1/3} \left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}-3 d x\right)^p dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-27*a*d^2+3*Sqrt[3]*d*Sqrt[4*b^3*d+27*a^2*d^2],3]},
  (a+b*x+d*x^3)^p/(((6*b*d-2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      ((6*(1+I*Sqrt[3])*b*d-2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p*
      ((6*(1-I*Sqrt[3])*b*d-2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p)*
    Int[(e+f*x)^m*((6*b*d-2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      ((6*(1+I*Sqrt[3])*b*d-2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p*
      ((6*(1-I*Sqrt[3])*b*d-2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)-3*d*x)^p,x]] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[p]] && NeQ[4*b^3+27*a^2*d,0]
```

2. $\int u \left(a + c x^2 + d x^3\right)^p dx$

1. $\int \left(a + c x^2 + d x^3\right)^p dx$

1. $\int \left(a + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z}$

1: $\int \left(a + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z} \wedge 4 c^3 + 27 a d^2 == 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $4 c^3 + 27 a d^2 == 0$**, then** $a + c x^2 + d x^3 == -\frac{1}{3^3 d^2} (c - 3 d x) (2 c + 3 d x)^2$
- **Rule: If** $p \in \mathbb{Z} \wedge 4 c^3 + 27 a d^2 == 0$**, then**

$$\int \left(a + c x^2 + d x^3\right)^p dx \longrightarrow -\frac{1}{3^{3 p} d^{2 p}} \int (c - 3 d x)^p (2 c + 3 d x)^{2 p} dx$$

- **Program code:**

```
Int[(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  -1/(3^(3*p)*d^(2*p))*Int[(c-3*d*x)^p*(2*c+3*d*x)^(2*p),x] /;
FreeQ[{a,c,d},x] && IntegerQ[p] && EqQ[4*c^3+27*a*d^2,0]
```

2. $\int \left(a + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z} \wedge 4 c^3 + 27 a d^2 \neq 0$

1: $\int \left(a + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z}^+ \wedge 4 c^3 + 27 a d^2 \neq 0$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^+ \wedge 4 c^3 + 27 a d^2 \neq 0$**,**

$$\int \left(a + c x^2 + d x^3\right)^p dx \longrightarrow \int \text{ExpandToSum}\left[\left(a + c x^2 + d x^3\right)^p, x\right] dx$$

- **Program code:**

```
Int[(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  Int[ExpandToSum[(a+c*x^2+d*x^3)^p,x],x] /;
FreeQ[{a,c,d},x] && IGtQ[p,0] && NeQ[4*c^3+27*a*d^2,0]
```

2. $\int \left(a + c\, x^2 + d\, x^3\right)^p \, dx$ when $p \in \mathbb{Z}^- \wedge 4\, c^3 + 27\, a\, d^2 \neq 0$

1: $\int \left(a + c\, x^2 + d\, x^3\right)^p \, dx$ when $p \in \mathbb{Z}^- \wedge 4\, c^3 + 27\, a\, d^2 \neq 0 \wedge \text{Factor}\left[a + c\, x^2 + d\, x^3\right] == v\, w \cdots$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^- \wedge 4\, c^3 + 27\, a\, d^2 \neq 0 \wedge \text{Factor}\left[a + c\, x^2 + d\, x^3\right] == v\, w \cdots$**, then**

$$\int \left(a + c\, x^2 + d\, x^3\right)^p \, dx \longrightarrow \int v^p\, w^p \cdots \, dx$$

- **Program code:**

```
Int[(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=Factor[a+c*x^2+d*x^3]},
  FreeFactors[u,x]^p*Int[DistributeDegree[NonfreeFactors[u,x],p],x] /;
 ProductQ[NonfreeFactors[u,x]]] /;
FreeQ[{a,c,d},x] && ILtQ[p,0] && NeQ[4*c^3+27*a*d^2,0]
```

2: $\int \left(a + c x^2 + d x^3\right)^p \, dx$ when $p \in \mathbb{Z}^- \land 4 c^3 + 27 a d^2 \neq 0$

- **Derivation: Algebraic expansion**

- **Basis: If** $r = \left(-2 c^3 - 27 a d^2 + 3 \sqrt{3} \, d \sqrt{4 a c^3 + 27 a^2 d^2}\right)^{1/3}$**, then**

$$a + c x^2 + d x^3 == \frac{1}{3^3 d^2} \left(c - \frac{2 c^2 + 2^{1/3} r^2}{2^{2/3} r} + 3 d x\right) \left(c + \frac{2 \left(1 + i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 - i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right) \left(c + \frac{2 \left(1 - i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 + i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)$$

- **Rule: If** $p \in \mathbb{Z}^- \land 4 c^3 + 27 a d^2 \neq 0$**, let** $r = \left(-2 c^3 - 27 a d^2 + 3 \sqrt{3} \, d \sqrt{4 a c^3 + 27 a^2 d^2}\right)^{1/3}$**, then**

$$\int \left(a + c x^2 + d x^3\right)^p \, dx \longrightarrow$$

$$\frac{1}{3^{3 p} d^{2 p}} \int \left(c - \frac{2 c^2 + 2^{1/3} r^2}{2^{2/3} r} + 3 d x\right)^p \left(c + \frac{2 \left(1 + i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 - i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)^p \left(c + \frac{2 \left(1 - i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 + i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)^p dx$$

- **Program code:**

```
Int[(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-2*c^3-27*a*d^2+3*Sqrt[3]*d*Sqrt[4*a*c^3+27*a^2*d^2],3]},
  1/(3^(3*p)*d^(2*p))*
    Int[(c-(2*c^2+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2+2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2+2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p,x]] /;
FreeQ[{a,c,d},x] && ILtQ[p,0] && NeQ[4*c^3+27*a*d^2,0]
```

2. $\int (a + c x^2 + d x^3)^p \, dx$ when $p \notin \mathbb{Z}$

1: $\int (a + c x^2 + d x^3)^p \, dx$ when $p \notin \mathbb{Z} \land 4 c^3 + 27 a d^2 == 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $4 c^3 + 27 a d^2 == 0$**, then** $\partial_x \frac{(a+c x^2+d x^3)^p}{(c-3 d x)^p (2 c+3 d x)^{2 p}} == 0$
- **Rule: If** $p \notin \mathbb{Z} \land 4 c^3 + 27 a d^2 == 0$**, then**

$$\int (a + c x^2 + d x^3)^p \, dx \to \frac{(a + c x^2 + d x^3)^p}{(c - 3 d x)^p (2 c + 3 d x)^{2 p}} \int (c - 3 d x)^p (2 c + 3 d x)^{2 p} \, dx$$

- **Program code:**

```
Int[(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  (a+c*x^2+d*x^3)^p/((c-3*d*x)^p*(2*c+3*d*x)^(2*p))*Int[(c-3*d*x)^p*(2*c+3*d*x)^(2*p),x] /;
FreeQ[{a,c,d,p},x] && Not[IntegerQ[p]] && EqQ[4*c^3+27*a*d^2,0]
```

2. $\int (a + c x^2 + d x^3)^p \, dx$ when $p \notin \mathbb{Z} \land 4 c^3 + 27 a d^2 \neq 0$

1: $\int (a + c x^2 + d x^3)^p \, dx$ when $p \notin \mathbb{Z} \land 4 c^3 + 27 a d^2 \neq 0 \land \text{Factor}[a + c x^2 + d x^3] == v w \cdots$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $\text{Factor}[a + c x^2 + d x^3] == v w \cdots$**, then** $\partial_x \frac{(a+c x^2+d x^3)^p}{v^p w^p \cdots} == 0$
- **Rule: If** $p \notin \mathbb{Z} \land 4 c^3 + 27 a d^2 \neq 0 \land \text{Factor}[a + c x^2 + d x^3] == v w \cdots$**, then**

$$\int (a + c x^2 + d x^3)^p \, dx \to \frac{(a + c x^2 + d x^3)^p}{v^p w^p \cdots} \int v^p w^p \cdots \, dx$$

- **Program code:**

```
Int[(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=NonfreeFactors[Factor[a+c*x^2+d*x^3],x]},
  (a+c*x^2+d*x^3)^p/DistributeDegree[u,p]*Int[DistributeDegree[u,p],x] /;
 ProductQ[u]] /;
FreeQ[{a,c,d,p},x] && Not[IntegerQ[p]] && NeQ[4*c^3+27*a*d^2,0]
```

2: $\int \left(a + c x^2 + d x^3\right)^p dx$ when $p \notin \mathbb{Z} \land 4 c^3 + 27 a d^2 \neq 0$

- **Derivation: Piecewise constant extraction**

- **Basis: If** $r = \left(-2 c^3 - 27 a d^2 + 3 \sqrt{3}\, d \sqrt{4 a c^3 + 27 a^2 d^2}\right)^{1/3}$**, then**

$$\partial_x \left(a + c x^2 + d x^3\right)^p \Big/ \left(\left(c - \frac{2 c^2 + 2^{1/3} r^2}{2^{2/3} r} + 3 d x\right)^p \left(c + \frac{2 \left(1 + i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 - i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)^p \left(c + \frac{2 \left(1 - i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 + i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)^p\right) == 0$$

- **Rule: If** $p \notin \mathbb{Z} \land 4 c^3 + 27 a d^2 \neq 0$**, let** $r = \left(-2 c^3 - 27 a d^2 + 3 \sqrt{3}\, d \sqrt{4 a c^3 + 27 a^2 d^2}\right)^{1/3}$**, then**

$$\int \left(a + c x^2 + d x^3\right)^p dx \longrightarrow$$

$$\left(a + c x^2 + d x^3\right)^p \Big/ \left(\left(c - \frac{2 c^2 + 2^{1/3} r^2}{2^{2/3} r} + 3 d x\right)^p \left(c + \frac{2 \left(1 + i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 - i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)^p \left(c + \frac{2 \left(1 - i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 + i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)^p\right) \cdot$$

$$\int \left(c - \frac{2 c^2 + 2^{1/3} r^2}{2^{2/3} r} + 3 d x\right)^p \left(c + \frac{2 \left(1 + i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 - i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)^p \left(c + \frac{2 \left(1 - i \sqrt{3}\right) c^2 + 2^{1/3} \left(1 + i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r} + 3 d x\right)^p dx$$

- **Program code:**

```
Int[(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-2*c^3-27*a*d^2+3*Sqrt[3]*d*Sqrt[4*a*c^3+27*a^2*d^2],3]},
  (a+c*x^2+d*x^3)^p/((c-(2*c^2+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2+2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2+2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p)*
    Int[(c-(2*c^2+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2+2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2+2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p,x]] /;
FreeQ[{a,c,d,p},x] && Not[IntegerQ[p]] && NeQ[4*c^3+27*a*d^2,0]
```

2. $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$

1. $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z}$

1: $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z} \wedge 4 c^3+27 a d^2 = 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $4 c^3+27 a d^2 = 0$**, then** $a+c x^2+d x^3 = -\frac{1}{3^3 d^2} (c-3 d x) (2 c+3 d x)^2$
- **Rule: If** $p \in \mathbb{Z} \wedge 4 c^3+27 a d^2 = 0$**, then**

$$\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx \rightarrow -\frac{1}{3^{3 p} d^{2 p}} \int (e+f x)^m (c-3 d x)^p (2 c+3 d x)^{2 p} \, dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  -1/(3^(3*p)*d^(2*p))*Int[(e+f*x)^m*(c-3*d*x)^p*(2*c+3*d*x)^(2*p),x] /;
FreeQ[{a,c,d,e,f,m},x] && IntegerQ[p] && EqQ[4*c^3+27*a*d^2,0]
```

2. $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z} \wedge 4 c^3+27 a d^2 \neq 0$

1: $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z}^+ \wedge 4 c^3+27 a d^2 \neq 0$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^+ \wedge 4 c^3+27 a d^2 \neq 0$**,**

$$\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx \rightarrow \int \text{ExpandIntegrand}\left[(e+f x)^m \left(a+c x^2+d x^3\right)^p, x\right] \, dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(e+f*x)^m*(a+c*x^2+d*x^3)^p,x],x] /;
FreeQ[{a,c,d,e,f,m},x] && IGtQ[p,0] && NeQ[4*c^3+27*a*d^2,0]
```

2. $\int (e + f x)^m \left(a + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z}^- \land 4 c^3 + 27 a d^2 \neq 0$

1: $\int (e + f x)^m \left(a + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z}^- \land 4 c^3 + 27 a d^2 \neq 0 \land \text{Factor}\left[a + c x^2 + d x^3\right] == v\, w \cdots$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^- \land 4 c^3 + 27 a d^2 \neq 0 \land \text{Factor}\left[a + c x^2 + d x^3\right] == v\, w \cdots$**, then**

$$\int (e + f x)^m \left(a + c x^2 + d x^3\right)^p dx \longrightarrow \int (e + f x)^m v^p w^p \cdots dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=Factor[a+c*x^2+d*x^3]},
  FreeFactors[u,x]^p*Int[(e+f*x)^m*DistributeDegree[NonfreeFactors[u,x],p],x] /;
 ProductQ[NonfreeFactors[u,x]]] /;
FreeQ[{a,c,d,e,f,m},x] && ILtQ[p,0] && NeQ[4*c^3+27*a*d^2,0]
```

2: $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p dx$ when $p \in \mathbb{Z}^- \land 4 c^3+27 a d^2 \neq 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $r=\left(-2 c^3-27 a d^2+3 \sqrt{3}\, d \sqrt{4 a c^3+27 a^2 d^2}\right)^{1/3}$**, then**

$$a+c x^2+d x^3 = \frac{1}{3^3 d^2}\left(c-\frac{2 c^2+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)\left(c+\frac{2\left(1+i \sqrt{3}\right) c^2+2^{1/3}\left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)\left(c+\frac{2\left(1-i \sqrt{3}\right) c^2+2^{1/3}\left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)$$

- **Rule: If** $p \in \mathbb{Z}^- \land 4 c^3+27 a d^2 \neq 0$**, let** $r=\left(-2 c^3-27 a d^2+3 \sqrt{3}\, d \sqrt{4 a c^3+27 a^2 d^2}\right)^{1/3}$**, then**

$$\int (e+f x)^m \left(a+c x^2+d x^3\right)^p dx \longrightarrow$$

$$\frac{1}{3^{3 p} d^{2 p}} \int (e+f x)^m \left(c-\frac{2 c^2+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1+i \sqrt{3}\right) c^2+2^{1/3}\left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1-i \sqrt{3}\right) c^2+2^{1/3}\left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-2*c^3-27*a*d^2+3*Sqrt[3]*d*Sqrt[4*a*c^3+27*a^2*d^2],3]},
  1/(3^(3*p)*d^(2*p))*
    Int[(e+f*x)^m*(c-(2*c^2+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2+2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2+2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p,x]] /;
FreeQ[{a,c,d,e,f,m},x] && ILtQ[p,0] && NeQ[4*c^3+27*a*d^2,0]
```

2. $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z}$

1: $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 c^3+27 a d^2 == 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If $4 c^3+27 a d^2 == 0$, then $\partial_x \frac{(a+c x^2+d x^3)^p}{(c-3 d x)^p (2 c+3 d x)^{2 p}} == 0$**
- **Rule: If $p \notin \mathbb{Z} \wedge 4 c^3+27 a d^2 == 0$, then**

$$\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx \longrightarrow \frac{\left(a+c x^2+d x^3\right)^p}{(c-3 d x)^p (2 c+3 d x)^{2 p}} \int (e+f x)^m (c-3 d x)^p (2 c+3 d x)^{2 p} \, dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  (a+c*x^2+d*x^3)^p/((c-3*d*x)^p*(2*c+3*d*x)^(2*p))*Int[(e+f*x)^m*(c-3*d*x)^p*(2*c+3*d*x)^(2*p),x] /;
FreeQ[{a,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && EqQ[4*c^3+27*a*d^2,0]
```

2. $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 c^3+27 a d^2 \neq 0$

1: $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 c^3+27 a d^2 \neq 0 \wedge \text{Factor}\left[a+c x^2+d x^3\right] == v\, w \cdots$

- **Derivation: Piecewise constant extraction**
- **Basis: If $\text{Factor}\left[a+c x^2+d x^3\right] == v\, w \cdots$, then $\partial_x \frac{(a+c x^2+d x^3)^p}{v^p w^p \cdots} == 0$**
- **Rule: If $p \notin \mathbb{Z} \wedge 4 c^3+27 a d^2 \neq 0 \wedge \text{Factor}\left[a+c x^2+d x^3\right] == v\, w \cdots$, then**

$$\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx \longrightarrow \frac{\left(a+c x^2+d x^3\right)^p}{v^p w^p \cdots} \int (e+f x)^m v^p w^p \cdots \, dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=NonfreeFactors[Factor[a+c*x^2+d*x^3],x]},
  (a+c*x^2+d*x^3)^p/DistributeDegree[u,p]*Int[(e+f*x)^m*DistributeDegree[u,p],x] /;
 ProductQ[u]] /;
FreeQ[{a,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && NeQ[4*c^3+27*a*d^2,0]
```

2: $\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge 4 c^3+27 a d^2 \neq 0$

- **Derivation: Piecewise constant extraction**

- **Basis: If** $r=\left(-2 c^3-27 a d^2+3 \sqrt{3}\, d \sqrt{4 a c^3+27 a^2 d^2}\right)^{1/3}$**, then**

$$\partial_x \left(a+c x^2+d x^3\right)^p \Big/ \left(\left(c-\frac{2 c^2+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1+i \sqrt{3}\right) c^2+2^{1/3}\left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1-i \sqrt{3}\right) c^2+2^{1/3}\left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p\right) == 0$$

- **Rule: If** $p \notin \mathbb{Z} \wedge 4 c^3+27 a d^2 \neq 0$**, let** $r=\left(-2 c^3-27 a d^2+3 \sqrt{3}\, d \sqrt{4 a c^3+27 a^2 d^2}\right)^{1/3}$**, then**

$$\int (e+f x)^m \left(a+c x^2+d x^3\right)^p \, dx \longrightarrow$$

$$\left(a+c x^2+d x^3\right)^p \Big/ \left(\left(c-\frac{2 c^2+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1+i \sqrt{3}\right) c^2+2^{1/3}\left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1-i \sqrt{3}\right) c^2+2^{1/3}\left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p\right) \cdot$$

$$\int (e+f x)^m \left(c-\frac{2 c^2+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1+i \sqrt{3}\right) c^2+2^{1/3}\left(1-i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1-i \sqrt{3}\right) c^2+2^{1/3}\left(1+i \sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-2*c^3-27*a*d^2+3*Sqrt[3]*d*Sqrt[4*a*c^3+27*a^2*d^2],3]},
  (a+c*x^2+d*x^3)^p/((c-(2*c^2+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2+2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2+2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p)*
    Int[(e+f*x)^m*(c-(2*c^2+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2+2^(1/3)*(1-I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2+2^(1/3)*(1+I*Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p,x]] /;
FreeQ[{a,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && NeQ[4*c^3+27*a*d^2,0]
```

3. $\int u \left(a+b x+c x^2+d x^3\right)^p dx$

1. $\int \left(a+b x+c x^2+d x^3\right)^p dx$

1. $\int \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \in \mathbb{Z}$

1. $\int \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \in \mathbb{Z} \land c^2-3 b d == 0$

1: $\int \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \in \mathbb{Z} \land c^2-3 b d == 0 \land b^2-3 a c == 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $c^2-3 b d == 0 \land b^2-3 a c == 0$**, then** $a+b x+c x^2+d x^3 == \frac{1}{3 b c} (b+c x)^3$
- **Rule: If** $p \in \mathbb{Z} \land c^2-3 b d == 0 \land b^2-3 a c == 0$**, then**

$$\int \left(a+b x+c x^2+d x^3\right)^p dx \longrightarrow \frac{1}{3^p b^p c^p} \int (b+c x)^{3 p} dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^p*b^p*c^p)*Int[(b+c*x)^(3*p),x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[p] && EqQ[c^2-3*b*d,0] && EqQ[b^2-3*a*c,0]
```

2: $\int \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \in \mathbb{Z} \land c^2-3 b d == 0 \land b^2-3 a c \neq 0$

- **Derivation: Integration by substitution**
- **Basis: If** $p \in \mathbb{Z} \land c^2-3 b d == 0$**, then** $\left(a+b x+c x^2+d x^3\right)^p == \frac{1}{3^p b^p c^p} \text{Subst}\left[\left(3 a b c-b^3+c^3 x^3\right)^p, x, x+\frac{c}{3 d}\right] \partial_x \left(x+\frac{c}{3 d}\right)$
- **Rule: If** $p \in \mathbb{Z} \land c^2-3 b d == 0 \land b^2-3 a c \neq 0$**, then**

$$\int \left(a+b x+c x^2+d x^3\right)^p dx \longrightarrow \frac{1}{3^p b^p c^p} \text{Subst}\left[\int \left(3 a b c-b^3+c^3 x^3\right)^p dx, x, x+\frac{c}{3 d}\right]$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^p*b^p*c^p)*Subst[Int[(3*a*b*c-b^3+c^3*x^3)^p,x],x,x+c/(3*d)] /;
FreeQ[{a,b,c,d},x] && IntegerQ[p] && EqQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

x: $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \in \mathbb{Z} \wedge c^2-3 b d=0 \wedge b^2-3 a c \neq 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $c^2-3 b d=0$, **let** $r=\left(b^3-3 a b c\right)^{1/3}$, **then** $a+b x+c x^2+d x^3=\frac{1}{3 b c}(b-r+c x)\left(b+\frac{\left(1-i \sqrt{3}\right) r}{2}+c x\right)\left(b+\frac{\left(1+i \sqrt{3}\right) r}{2}+c x\right)$
- **Rule: If** $p \in \mathbb{Z} \wedge c^2-3 b d=0 \wedge b^2-3 a c \neq 0$, **let** $r=\left(b^3-3 a b c\right)^{1/3}$, **then**

$$\int (a+b x+c x^2+d x^3)^p \, dx \rightarrow \frac{1}{3^p b^p c^p} \int (b-r+c x)^p \left(b+\frac{\left(1-i \sqrt{3}\right) r}{2}+c x\right)^p \left(b+\frac{\left(1+i \sqrt{3}\right) r}{2}+c x\right)^p dx$$

- **Program code:**

```
(* Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[b^3-3*a*b*c,3]},
  1/(3^p*b^p*c^p)*Int[(b-r+c*x)^p*(b+(1-I*Sqrt[3])*r/2+c*x)^p*(b+(1+I*Sqrt[3])*r/2+c*x)^p,x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[p] && EqQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0] *)
```

2. $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \in \mathbb{Z} \wedge c^2-3 b d \neq 0$

1: $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \in \mathbb{Z} \wedge c^2-3 b d \neq 0 \wedge b^2-3 a c=0$

- **Derivation: Algebraic expansion**
- **Basis: If** $b^2-3 a c=0$, **let** $r=\left(c^3-3 b c d\right)^{1/3}$, **then** $a+b x+c x^2+d x^3=\frac{1}{3 b c}(b+(c-r) x)\left(b+\left(c+\frac{\left(1-i \sqrt{3}\right) r}{2}\right) x\right)\left(b+\left(c+\frac{\left(1+i \sqrt{3}\right) r}{2}\right) x\right)$
- **Rule: If** $p \in \mathbb{Z} \wedge c^2-3 b d \neq 0 \wedge b^2-3 a c=0$, **let** $r=\left(c^3-3 b c d\right)^{1/3}$, **then**

$$\int (a+b x+c x^2+d x^3)^p \, dx \rightarrow \frac{1}{3^p b^p c^p} \int (b+(c-r) x)^p \left(b+\left(c+\frac{\left(1-i \sqrt{3}\right) r}{2}\right) x\right)^p \left(b+\left(c+\frac{\left(1+i \sqrt{3}\right) r}{2}\right) x\right)^p dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[c^3-3*b*c*d,3]},
  1/(3^p*b^p*c^p)*Int[(b+(c-r)*x)^p*(b+(c+(1-I*Sqrt[3])*r/2)*x)^p*(b+(c+(1+I*Sqrt[3])*r/2)*x)^p,x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[p] && NeQ[c^2-3*b*d,0] && EqQ[b^2-3*a*c,0]
```

2. $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \in \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

1: $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \in \mathbb{Z}^+ \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^+ \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$,

$$\int (a+b x+c x^2+d x^3)^p \, dx \longrightarrow \int \text{ExpandToSum}\left[(a+b x+c x^2+d x^3)^p, x\right] dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  Int[ExpandToSum[(a+b*x+c*x^2+d*x^3)^p,x],x] /;
FreeQ[{a,b,c,d},x] && IGtQ[p,0] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

2. $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

1: $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0 \land \text{Factor}\left[a+b x+c x^2+d x^3\right] == v\, w \cdots$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0 \land \text{Factor}\left[a+b x+c x^2+d x^3\right] == v\, w \cdots$, **then**

$$\int (a+b x+c x^2+d x^3)^p \, dx \longrightarrow \int v^p\, w^p \cdots dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=Factor[a+b*x+c*x^2+d*x^3]},
  FreeFactors[u,x]^p*Int[DistributeDegree[NonfreeFactors[u,x],p],x] /;
 ProductQ[NonfreeFactors[u,x]]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

2: $\int \left(a+b\,x+c\,x^2+d\,x^3\right)^p\,dx$ when $p\in\mathbb{Z}^-\land c^2-3\,b\,d\neq 0\land b^2-3\,a\,c\neq 0$

- **Derivation: Integration by substitution**
- **Basis: If** $p\in\mathbb{Z}$, **then** $\left(a+b\,x+c\,x^2+d\,x^3\right)^p == \frac{1}{3^{3\,p}\,d^{2\,p}}\,\text{Subst}\left[\left(2\,c^3-9\,b\,c\,d+27\,a\,d^2-9\,d\,\left(c^2-3\,b\,d\right)\,x+27\,d^3\,x^3\right)^p,\,x,\,x+\frac{c}{3\,d}\right]\,\partial_x\left(x+\frac{c}{3\,d}\right)$
- **Note: The substitution transforms the cubic polynomial in the integrand into a reduced cubic of the form** $r+s\,x+t\,x^3$.
- **Rule: If** $p\in\mathbb{Z}^-\land c^2-3\,b\,d\neq 0\land b^2-3\,a\,c\neq 0$, **then**

$$\int \left(a+b\,x+c\,x^2+d\,x^3\right)^p\,dx\;\longrightarrow\;\frac{1}{3^{3\,p}\,d^{2\,p}}\,\text{Subst}\left[\int \left(2\,c^3-9\,b\,c\,d+27\,a\,d^2-9\,d\,\left(c^2-3\,b\,d\right)\,x+27\,d^3\,x^3\right)^p\,dx,\,x,\,x+\frac{c}{3\,d}\right]$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^(3*p)*d^(2*p))*Subst[Int[(2*c^3-9*b*c*d+27*a*d^2-9*d*(c^2-3*b*d)*x+27*d^3*x^3)^p,x],x,x+c/(3*d)] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

x: $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $r = \left(-2 c^3+9 b c d-27 a d^2+3 \sqrt{3}\, d \sqrt{-b^2 c^2+4 a c^3+4 b^3 d-18 a b c d+27 a^2 d^2}\right)^{1/3}$**, then**

$$a+b x+c x^2+d x^3 == \frac{1}{27 d^2} \left(c-\frac{2 c^2-6 b d+2^{1/3} r^2}{2^{2/3} r}+3 d x\right) \left(c+\frac{2 \left(1+i \sqrt{3}\right) c^2-6 \left(1+i \sqrt{3}\right) b d-i\, 2^{1/3} \left(i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right) \left(c+\frac{2 \left(1-i \sqrt{3}\right) c^2-6 \left(1-i \sqrt{3}\right) b d+i\, 2^{1/3} \left(-i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)$$

- **Rule: If** $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$**, let** $r = \left(-2 c^3+9 b c d-27 a d^2+3 \sqrt{3}\, d \sqrt{-b^2 c^2+4 a c^3+4 b^3 d-18 a b c d+27 a^2 d^2}\right)^{1/3}$**, then**

$$\int (a+b x+c x^2+d x^3)^p \, dx \longrightarrow$$

$$\frac{1}{3^{3 p} d^{2 p}} \int \left(c-\frac{2 c^2-6 b d+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p$$

$$\left(c+\frac{2 \left(1+i \sqrt{3}\right) c^2-6 \left(1+i \sqrt{3}\right) b d-i\, 2^{1/3} \left(i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2 \left(1-i \sqrt{3}\right) c^2-6 \left(1-i \sqrt{3}\right) b d+i\, 2^{1/3} \left(-i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p dx$$

- **Program code:**

```
(* Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-2*c^3+9*b*c*d-27*a*d^2+3*Sqrt[3]*d*Sqrt[-b^2*c^2+4*a*c^3+4*b^3*d-18*a*b*c*d+27*a^2*d^2],3]},
  1/(3^(3*p)*d^(2*p))*
    Int[(c-(2*c^2-6*b*d+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2-6*(1+I*Sqrt[3])*b*d-I*2^(1/3)*(I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2-6*(1-I*Sqrt[3])*b*d+I*2^(1/3)*(-I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p,x]] /;
FreeQ[{a,b,c,d},x] && ILtQ[p,0] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0] *)
```

2. $\int \left(a + b x + c x^2 + d x^3\right)^p \, dx$ when $p \notin \mathbb{Z}$

1. $\int \left(a + b x + c x^2 + d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge c^2 - 3 b d == 0$

1: $\int \left(a + b x + c x^2 + d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \wedge c^2 - 3 b d == 0 \wedge b^2 - 3 a c == 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $c^2 - 3 b d == 0 \wedge b^2 - 3 a c == 0$**, then** $\partial_x \frac{\left(a+b x+c x^2+d x^3\right)^p}{(b+c x)^{3 p}} == 0$
- **Rule: If** $p \notin \mathbb{Z} \wedge c^2 - 3 b d == 0 \wedge b^2 - 3 a c == 0$**, then**

$$\int \left(a + b x + c x^2 + d x^3\right)^p \, dx \longrightarrow \frac{\left(a + b x + c x^2 + d x^3\right)^p}{(b + c x)^{3 p}} \int (b + c x)^{3 p} \, dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  (a+b*x+c*x^2+d*x^3)^p/(b+c*x)^(3*p)*Int[(b+c*x)^(3*p),x] /;
FreeQ[{a,b,c,d,p},x] && Not[IntegerQ[p]] && EqQ[c^2-3*b*d,0] && EqQ[b^2-3*a*c,0]
```

2: $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \notin \mathbb{Z} \land c^2-3 b d=0 \land b^2-3 a c \neq 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $c^2-3 b d=0$, **let** $r=\left(b^3-3 a b c\right)^{1/3}$, **then** $\partial_x \frac{(a+b x+c x^2+d x^3)^p}{(b-r+c x)^p \left(b+\frac{(1-i\sqrt{3}) r}{2}+c x\right)^p \left(b+\frac{(1+i\sqrt{3}) r}{2}+c x\right)^p}=0$
- **Rule: If** $p \notin \mathbb{Z} \land c^2-3 b d=0 \land b^2-3 a c \neq 0$, **let** $r=\left(b^3-3 a b c\right)^{1/3}$, **then**

$$\int (a+b x+c x^2+d x^3)^p \, dx \to$$

$$\frac{(a+b x+c x^2+d x^3)^p}{(b-r+c x)^p \left(b+\frac{(1-i\sqrt{3}) r}{2}+c x\right)^p \left(b+\frac{(1+i\sqrt{3}) r}{2}+c x\right)^p} \int (b-r+c x)^p \left(b+\frac{(1-i\sqrt{3}) r}{2}+c x\right)^p \left(b+\frac{(1+i\sqrt{3}) r}{2}+c x\right)^p dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[b^3-3*a*b*c,3]},
  (a+b*x+c*x^2+d*x^3)^p/((b-r+c*x)^p*(b+(1-I*Sqrt[3])*r/2+c*x)^p*(b+(1+I*Sqrt[3])*r/2+c*x)^p)*
    Int[(b-r+c*x)^p*(b+(1-I*Sqrt[3])*r/2+c*x)^p*(b+(1+I*Sqrt[3])*r/2+c*x)^p,x]] /;
FreeQ[{a,b,c,d,p},x] && Not[IntegerQ[p]] && EqQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

2. $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \notin \mathbb{Z} \land c^2-3 b d \neq 0$

1: $\int (a+b x+c x^2+d x^3)^p \, dx$ when $p \notin \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c=0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $b^2-3 a c=0$, **let** $r=\left(c^3-3 b c d\right)^{1/3}$, **then** $\partial_x \frac{(a+b x+c x^2+d x^3)^p}{(b+(c-r) x)^p \left(b+\left(c+\frac{(1-i\sqrt{3}) r}{2}\right) x\right)^p \left(b+\left(c+\frac{(1+i\sqrt{3}) r}{2}\right) x\right)^p}=0$
- **Rule: If** $p \notin \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c=0$, **let** $r=\left(c^3-3 b c d\right)^{1/3}$, **then**

$$\int (a+b x+c x^2+d x^3)^p \, dx \to$$

$$\frac{\left(a+b x+c x^2+d x^3\right)^p}{(b+(c-r) x)^p\left(b+\left(c+\frac{\left(1-i \sqrt{3}\right) r}{2}\right) x\right)^p\left(b+\left(c+\frac{\left(1+i \sqrt{3}\right) r}{2}\right) x\right)^p}\int(b+(c-r) x)^p\left(b+\left(c+\frac{\left(1-i \sqrt{3}\right) r}{2}\right) x\right)^p\left(b+\left(c+\frac{\left(1+i \sqrt{3}\right) r}{2}\right) x\right)^p dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[c^3-3*b*c*d,3]},
  (a+b*x+c*x^2+d*x^3)^p/((b+(c-r)*x)^p*(b+(c+(1-I*Sqrt[3])*r/2)*x)^p*(b+(c+(1+I*Sqrt[3])*r/2)*x)^p)*
    Int[(b+(c-r)*x)^p*(b+(c+(1-I*Sqrt[3])*r/2)*x)^p*(b+(c+(1+I*Sqrt[3])*r/2)*x)^p,x]] /;
FreeQ[{a,b,c,d,p},x] && Not[IntegerQ[p]] && NeQ[c^2-3*b*d,0] && EqQ[b^2-3*a*c,0]
```

2. $\int\left(a+b x+c x^2+d x^3\right)^p dx$ when $p \notin \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

1: $\int\left(a+b x+c x^2+d x^3\right)^p dx$ when $p \notin \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0 \land \text{Factor}\left[a+b x+c x^2+d x^3\right] == v\, w \cdots$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $\text{Factor}\left[a+b x+c x^2+d x^3\right] == v\, w \cdots$, **then** $\partial_x \frac{\left(a+b x+c x^2+d x^3\right)^p}{v^p w^p \cdots} == 0$
- **Rule: If** $p \notin \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0 \land \text{Factor}\left[a+b x+c x^2+d x^3\right] == v\, w \cdots$, **then**

$$\int\left(a+b x+c x^2+d x^3\right)^p dx \longrightarrow \frac{\left(a+b x+c x^2+d x^3\right)^p}{v^p w^p \cdots}\int v^p w^p \cdots dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=NonfreeFactors[Factor[a+b*x+c*x^2+d*x^3],x]},
  (a+b*x+c*x^2+d*x^3)^p/DistributeDegree[u,p]*Int[DistributeDegree[u,p],x] /;
 ProductQ[u]] /;
FreeQ[{a,b,c,d,p},x] && Not[IntegerQ[p]] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

x: $\int \left(a + b\,x + c\,x^2 + d\,x^3\right)^p \, dx$ when $p \notin \mathbb{Z}$?? ??

- **Derivation: Integration by substitution**
- **Basis: If** $p \in \mathbb{Z}$**, then** $\left(a + b\,x + c\,x^2 + d\,x^3\right)^p == \frac{1}{3^{3\,p}\,d^{2\,p}}\,\text{Subst}\left[\left(2\,c^3 - 9\,b\,c\,d + 27\,a\,d^2 - 9\,d\,\left(c^2 - 3\,b\,d\right)\,x + 27\,d^3\,x^3\right)^p,\ x,\ x + \frac{c}{3\,d}\right]\,\partial_x\left(x + \frac{c}{3\,d}\right)$
- **Note: The substitution transforms the cubic polynomial in the integrand into a reduced cubic of the form** $r + s\,x + t\,x^3$.
- **Rule: If** $p \notin \mathbb{Z}$**, then**

$$\int \left(a + b\,x + c\,x^2 + d\,x^3\right)^p \, dx \ \longrightarrow\ \frac{1}{3^{3\,p}\,d^{2\,p}}\,\text{Subst}\left[\int \left(2\,c^3 - 9\,b\,c\,d + 27\,a\,d^2 - 9\,d\,\left(c^2 - 3\,b\,d\right)\,x + 27\,d^3\,x^3\right)^p \, dx,\ x,\ x + \frac{c}{3\,d}\right]$$

- **Program code:**

```
(* Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^(3*p)*d^(2*p))*Subst[Int[(2*c^3-9*b*c*d+27*a*d^2-9*d*(c^2-3*b*d)*x+27*d^3*x^3)^p,x],x,x+c/(3*d)] /;
FreeQ[{a,b,c,d,p},x] && Not[IntegerQ[p]] *)
```

2: $\int \left(a + b\,x + c\,x^2 + d\,x^3\right)^p \, dx$ when $p \notin \mathbb{Z}\ \land\ c^2 - 3\,b\,d \neq 0\ \land\ b^2 - 3\,a\,c \neq 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $r = \left(-2\,c^3 + 9\,b\,c\,d - 27\,a\,d^2 + 3\,\sqrt{3}\,d\,\sqrt{-b^2\,c^2 + 4\,a\,c^3 + 4\,b^3\,d - 18\,a\,b\,c\,d + 27\,a^2\,d^2}\right)^{1/3}$**, then**

$$\partial_x\left(a + b\,x + c\,x^2 + d\,x^3\right)^p \Big/ \left(\left(c - \frac{2\,c^2 - 6\,b\,d + 2^{1/3}\,r^2}{2^{2/3}\,r} + 3\,d\,x\right)^p \left(c + \frac{2\,\left(1 + i\,\sqrt{3}\right)\,c^2 - 6\,\left(1 + i\,\sqrt{3}\right)\,b\,d - i\,2^{1/3}\,\left(i + \sqrt{3}\right)\,r^2}{2 \times 2^{2/3}\,r} + 3\,d\,x\right)^p \left(c + \frac{2\,\left(1 - i\,\sqrt{3}\right)\,c^2 - 6\,\left(1 - i\,\sqrt{3}\right)\,b\,d + i\,2^{1/3}\,\left(-i + \sqrt{3}\right)\,r^2}{2 \times 2^{2/3}\,r} + 3\,d\,x\right)^p\right) == 0$$

- **Rule: If** $p \notin \mathbb{Z}\ \land\ c^2 - 3\,b\,d \neq 0\ \land\ b^2 - 3\,a\,c \neq 0$**, let** $r = \left(-2\,c^3 + 9\,b\,c\,d - 27\,a\,d^2 + 3\,\sqrt{3}\,d\,\sqrt{-b^2\,c^2 + 4\,a\,c^3 + 4\,b^3\,d - 18\,a\,b\,c\,d + 27\,a^2\,d^2}\right)^{1/3}$**, then**

$$\int \left(a + b\,x + c\,x^2 + d\,x^3\right)^p \, dx \ \longrightarrow$$

$$\left(a + b\,x + c\,x^2 + d\,x^3\right)^p \Big/ \left(\left(c - \frac{2\,c^2 - 6\,b\,d + 2^{1/3}\,r^2}{2^{2/3}\,r} + 3\,d\,x\right)^p\right.$$

$$\left(c+\frac{2\left(1+i\sqrt{3}\right)c^2-6\left(1+i\sqrt{3}\right)b\,d-i\,2^{1/3}\left(i+\sqrt{3}\right)r^2}{2\times 2^{2/3}\,r}+3\,d\,x\right)^p\left(c+\frac{2\left(1-i\sqrt{3}\right)c^2-6\left(1-i\sqrt{3}\right)b\,d+i\,2^{1/3}\left(-i+\sqrt{3}\right)r^2}{2\times 2^{2/3}\,r}+3\,d\,x\right)^p\bigg)\cdot$$

$$\int\left(c-\frac{2\,c^2-6\,b\,d+2^{1/3}\,r^2}{2^{2/3}\,r}+3\,d\,x\right)^p\left(c+\frac{2\left(1+i\sqrt{3}\right)c^2-6\left(1+i\sqrt{3}\right)b\,d-i\,2^{1/3}\left(i+\sqrt{3}\right)r^2}{2\times 2^{2/3}\,r}+3\,d\,x\right)^p$$
$$\left(c+\frac{2\left(1-i\sqrt{3}\right)c^2-6\left(1-i\sqrt{3}\right)b\,d+i\,2^{1/3}\left(-i+\sqrt{3}\right)r^2}{2\times 2^{2/3}\,r}+3\,d\,x\right)^p dx$$

- **Program code:**

```
Int[(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-2*c^3+9*b*c*d-27*a*d^2+3*Sqrt[3]*d*Sqrt[-b^2*c^2+4*a*c^3+4*b^3*d-18*a*b*c*d+27*a^2*d^2],3]},
  (a+b*x+c*x^2+d*x^3)^p/
    ((c-(2*c^2-6*b*d+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2-6*(1+I*Sqrt[3])*b*d-I*2^(1/3)*(I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2-6*(1-I*Sqrt[3])*b*d+I*2^(1/3)*(-I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p)*
    Int[(c-(2*c^2-6*b*d+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2-6*(1+I*Sqrt[3])*b*d-I*2^(1/3)*(I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2-6*(1-I*Sqrt[3])*b*d+I*2^(1/3)*(-I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p,x]] /;
FreeQ[{a,b,c,d,p},x] && Not[IntegerQ[p]] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

3: $\int u^p\,dx$ when $u == a+b\,x+c\,x^2+d\,x^3$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == a+b\,x+c\,x^2+d\,x^3$**, then**

$$\int u^p\,dx\;\longrightarrow\;\int\left(a+b\,x+c\,x^2+d\,x^3\right)^p dx$$

- **Program code:**

```
Int[u_^p_,x_Symbol] :=
  Int[ExpandToSum[u,x]^p,x] /;
FreeQ[p,x] && PolyQ[u,x,3] && Not[CubicMatchQ[u,x]]
```

2. $\int (e + f x)^m \left(a + b x + c x^2 + d x^3\right)^p dx$

1. $\int (e + f x)^m \left(a + b x + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z}$

1. $\int (e + f x)^m \left(a + b x + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z} \land c^2 - 3 b d == 0$

1: $\int (e + f x)^m \left(a + b x + c x^2 + d x^3\right)^p dx$ when $p \in \mathbb{Z} \land c^2 - 3 b d == 0 \land b^2 - 3 a c == 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $c^2 - 3 b d == 0 \land b^2 - 3 a c == 0$, **then** $a + b x + c x^2 + d x^3 == \frac{1}{3 b c} (b + c x)^3$
- **Rule: If** $p \in \mathbb{Z} \land c^2 - 3 b d == 0 \land b^2 - 3 a c == 0$, **then**

$$\int (e + f x)^m \left(a + b x + c x^2 + d x^3\right)^p dx \longrightarrow \frac{1}{3^p b^p c^p} \int (e + f x)^m (b + c x)^{3 p} dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^p*b^p*c^p)*Int[(e+f*x)^m*(b+c*x)^(3*p),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IntegerQ[p] && EqQ[c^2-3*b*d,0] && EqQ[b^2-3*a*c,0]
```

2: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z} \land c^2-3 b d = 0 \land b^2-3 a c \neq 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $c^2-3 b d = 0$, **let** $r = \left(b^3-3 a b c\right)^{1/3}$, **then** $a+b x+c x^2+d x^3 = \frac{1}{3 b c} (b-r+c x) \left(b+\frac{\left(1-i \sqrt{3}\right) r}{2}+c x\right) \left(b+\frac{\left(1+i \sqrt{3}\right) r}{2}+c x\right)$
- **Rule: If** $p \in \mathbb{Z} \land c^2-3 b d = 0 \land b^2-3 a c \neq 0$, **let** $r = \left(b^3-3 a b c\right)^{1/3}$, **then**

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx \to \frac{1}{3^p b^p c^p} \int (e+f x)^m (b-r+c x)^p \left(b+\frac{\left(1-i \sqrt{3}\right) r}{2}+c x\right)^p \left(b+\frac{\left(1+i \sqrt{3}\right) r}{2}+c x\right)^p dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[b^3-3*a*b*c,3]},
  1/(3^p*b^p*c^p)*Int[(e+f*x)^m*(b-r+c*x)^p*(b+(1-I*Sqrt[3])*r/2+c*x)^p*(b+(1+I*Sqrt[3])*r/2+c*x)^p,x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && IntegerQ[p] && EqQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

2. $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \in \mathbb{Z} \land c^2-3 b d \neq 0$

1: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \in \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c = 0$

- **Derivation: Algebraic expansion**
- **Basis: If** $b^2-3 a c = 0$, **let** $r = \left(c^3-3 b c d\right)^{1/3}$, **then** $a+b x+c x^2+d x^3 = \frac{1}{3 b c} (b+(c-r) x) \left(b+\left(c+\frac{\left(1-i \sqrt{3}\right) r}{2}\right) x\right) \left(b+\left(c+\frac{\left(1+i \sqrt{3}\right) r}{2}\right) x\right)$
- **Rule: If** $p \in \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c = 0$, **let** $r = \left(c^3-3 b c d\right)^{1/3}$, **then**

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx \to \frac{1}{3^p b^p c^p} \int (e+f x)^m (b+(c-r) x)^p \left(b+\left(c+\frac{\left(1-i \sqrt{3}\right) r}{2}\right) x\right)^p \left(b+\left(c+\frac{\left(1+i \sqrt{3}\right) r}{2}\right) x\right)^p dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[c^3-3*b*c*d,3]},
  1/(3^p*b^p*c^p)*Int[(e+f*x)^m*(b+(c-r)*x)^p*(b+(c+(1-I*Sqrt[3])*r/2)*x)^p*(b+(c+(1+I*Sqrt[3])*r/2)*x)^p,x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && IntegerQ[p] && NeQ[c^2-3*b*d,0] && EqQ[b^2-3*a*c,0]
```

2. $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

1: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z}^+ \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^+ \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$,

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(e+f x)^m \left(a+b x+c x^2+d x^3\right)^p, x\right] dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(e+f*x)^m*(a+b*x+c*x^2+d*x^3)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

2. $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

1: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0 \land \text{Factor}\left[a+b x+c x^2+d x^3\right] == v w \cdots$

- **Derivation: Algebraic expansion**
- **Rule: If** $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0 \land \text{Factor}\left[a+b x+c x^2+d x^3\right] == v w \cdots$**, then**

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx \longrightarrow \int (e+f x)^m v^p w^p \cdots \, dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=Factor[a+b*x+c*x^2+d*x^3]},
  FreeFactors[u,x]^p*Int[(e+f*x)^m*DistributeDegree[NonfreeFactors[u,x],p],x] /;
 ProductQ[NonfreeFactors[u,x]]] /;
FreeQ[{a,b,c,d,e,f,m},x] && ILtQ[p,0] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

2: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \in \mathbb{Z}^- \wedge c^2-3 b d \neq 0 \wedge b^2-3 a c \neq 0$?? ??

- **Derivation: Integration by substitution**
- **Basis: If** $p \in \mathbb{Z}$, **then** $\left(a+b x+c x^2+d x^3\right)^p == \frac{1}{3^{3 p} d^{2 p}} \text{Subst}\left[\left(2 c^3-9 b c d+27 a d^2-9 d \left(c^2-3 b d\right) x+27 d^3 x^3\right)^p, x, x+\frac{c}{3 d}\right] \partial_x \left(x+\frac{c}{3 d}\right)$
- **Note: The substitution transforms the cubic polynomial in the integrand into a reduced cubic of the form** $r+s x+t x^3$**.**
- **Rule: If** $p \in \mathbb{Z}^- \wedge c^2-3 b d \neq 0 \wedge b^2-3 a c \neq 0$**, then**

$$\int \left(a+b x+c x^2+d x^3\right)^p dx \longrightarrow \frac{1}{3^{3 p} d^{2 p}} \text{Subst}\left[\int \left(2 c^3-9 b c d+27 a d^2-9 d \left(c^2-3 b d\right) x+27 d^3 x^3\right)^p dx, x, x+\frac{c}{3 d}\right]$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^(3*p)*d^(2*p))*Subst[Int[(2*c^3-9*b*c*d+27*a*d^2-9*d*(c^2-3*b*d)*x+27*d^3*x^3)^p,x],x,x+c/(3*d)] /;
FreeQ[{a,b,c,d,e,f,m},x] && ILtQ[p,0] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

x: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

- **Derivation: Algebraic expansion**

- **Basis: If** $r = \left(-2 c^3+9 b c d-27 a d^2+3 \sqrt{3} d \sqrt{-b^2 c^2+4 a c^3+4 b^3 d-18 a b c d+27 a^2 d^2}\right)^{1/3}$**, then**

  $a+b x+c x^2+d x^3 = \frac{1}{27 d^2} \left(c-\frac{2 c^2-6 b d+2^{1/3} r^2}{2^{2/3} r}+3 d x\right) \left(c+\frac{2 \left(1+i \sqrt{3}\right) c^2-6 \left(1+i \sqrt{3}\right) b d-i\, 2^{1/3} \left(i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right) \left(c+\frac{2 \left(1-i \sqrt{3}\right) c^2-6 \left(1-i \sqrt{3}\right) b d+i\, 2^{1/3} \left(-i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)$

- **Rule: If** $p \in \mathbb{Z}^- \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$**, let** $r = \left(-2 c^3+9 b c d-27 a d^2+3 \sqrt{3} d \sqrt{-b^2 c^2+4 a c^3+4 b^3 d-18 a b c d+27 a^2 d^2}\right)^{1/3}$**, then**

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx \longrightarrow$$

$$\frac{1}{3^{3 p} d^{2 p}} \int (e+f x)^m \left(c-\frac{2 c^2-6 b d+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p$$

$$\left(c+\frac{2 \left(1+i \sqrt{3}\right) c^2-6 \left(1+i \sqrt{3}\right) b d-i\, 2^{1/3} \left(i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2 \left(1-i \sqrt{3}\right) c^2-6 \left(1-i \sqrt{3}\right) b d+i\, 2^{1/3} \left(-i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p dx$$

- **Program code:**

```
(* Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-2*c^3+9*b*c*d-27*a*d^2+3*Sqrt[3]*d*Sqrt[-b^2*c^2+4*a*c^3+4*b^3*d-18*a*b*c*d+27*a^2*d^2],3]},
  1/(3^(3*p)*d^(2*p))*
    Int[(e+f*x)^m*(c-(2*c^2-6*b*d+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2-6*(1+I*Sqrt[3])*b*d-I*2^(1/3)*(I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2-6*(1-I*Sqrt[3])*b*d+I*2^(1/3)*(-I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p,x]] /;
FreeQ[{a,b,c,d,e,f,m},x] && ILtQ[p,0] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0] *)
```

2. $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z}$

   1. $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \land c^2-3 b d == 0$

      1: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \land c^2-3 b d == 0 \land b^2-3 a c == 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $c^2-3 b d == 0 \land b^2-3 a c == 0$**, then** $\partial_x \frac{(a+b x+c x^2+d x^3)^p}{(b+c x)^{3 p}} == 0$
- **Rule: If** $p \notin \mathbb{Z} \land c^2-3 b d == 0 \land b^2-3 a c == 0$**, then**

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx \longrightarrow \frac{\left(a+b x+c x^2+d x^3\right)^p}{(b+c x)^{3 p}} \int (e+f x)^m (b+c x)^{3 p} \, dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  (a+b*x+c*x^2+d*x^3)^p/(b+c*x)^(3*p)*Int[(e+f*x)^m*(b+c*x)^(3*p),x] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && EqQ[c^2-3*b*d,0] && EqQ[b^2-3*a*c,0]
```

2: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \notin \mathbb{Z} \wedge c^2-3 b d == 0 \wedge b^2-3 a c \neq 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $c^2-3 b d == 0$, **let** $r = \left(b^3-3 a b c\right)^{1/3}$, **then** $\partial_x \frac{\left(a+b x+c x^2+d x^3\right)^p}{(b-r+c x)^p \left(b+\frac{\left(1-i \sqrt{3}\right) r}{2}+c x\right)^p \left(b+\frac{\left(1+i \sqrt{3}\right) r}{2}+c x\right)^p} == 0$
- **Rule: If** $p \notin \mathbb{Z} \wedge c^2-3 b d == 0 \wedge b^2-3 a c \neq 0$, **let** $r = \left(b^3-3 a b c\right)^{1/3}$, **then**

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx \longrightarrow$$

$$\frac{\left(a+b x+c x^2+d x^3\right)^p}{(b-r+c x)^p \left(b+\frac{\left(1-i \sqrt{3}\right) r}{2}+c x\right)^p \left(b+\frac{\left(1+i \sqrt{3}\right) r}{2}+c x\right)^p} \int (e+f x)^m (b-r+c x)^p \left(b+\frac{\left(1-i \sqrt{3}\right) r}{2}+c x\right)^p \left(b+\frac{\left(1+i \sqrt{3}\right) r}{2}+c x\right)^p dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[b^3-3*a*b*c,3]},
  (a+b*x+c*x^2+d*x^3)^p/((b-r+c*x)^p*(b+(1-I*Sqrt[3])*r/2+c*x)^p*(b+(1+I*Sqrt[3])*r/2+c*x)^p)*
    Int[(e+f*x)^m*(b-r+c*x)^p*(b+(1-I*Sqrt[3])*r/2+c*x)^p*(b+(1+I*Sqrt[3])*r/2+c*x)^p,x]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && EqQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

2. $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \notin \mathbb{Z} \wedge c^2-3 b d \neq 0$

1: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx$ when $p \notin \mathbb{Z} \wedge c^2-3 b d \neq 0 \wedge b^2-3 a c == 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $b^2-3 a c == 0$, **let** $r = \left(c^3-3 b c d\right)^{1/3}$, **then** $\partial_x \frac{\left(a+b x+c x^2+d x^3\right)^p}{(b+(c-r) x)^p \left(b+\left(c+\frac{\left(1-i \sqrt{3}\right) r}{2}\right) x\right)^p \left(b+\left(c+\frac{\left(1+i \sqrt{3}\right) r}{2}\right) x\right)^p} == 0$
- **Rule: If** $p \notin \mathbb{Z} \wedge c^2-3 b d \neq 0 \wedge b^2-3 a c == 0$, **let** $r = \left(c^3-3 b c d\right)^{1/3}$, **then**

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx \longrightarrow$$

$$\frac{\left(a+b\,x+c\,x^2+d\,x^3\right)^p}{\left(b+(c-r)\,x\right)^p\left(b+\left(c+\frac{\left(1-i\sqrt{3}\right)r}{2}\right)x\right)^p\left(b+\left(c+\frac{\left(1+i\sqrt{3}\right)r}{2}\right)x\right)^p}\int(e+f\,x)^m\,\left(b+(c-r)\,x\right)^p\left(b+\left(c+\frac{\left(1-i\sqrt{3}\right)r}{2}\right)x\right)^p\left(b+\left(c+\frac{\left(1+i\sqrt{3}\right)r}{2}\right)x\right)^p\,dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[c^3-3*b*c*d,3]},
  (a+b*x+c*x^2+d*x^3)^p/((b+(c-r)*x)^p*(b+(c+(1-I*Sqrt[3])*r/2)*x)^p*(b+(c+(1+I*Sqrt[3])*r/2)*x)^p)*
    Int[(e+f*x)^m*(b+(c-r)*x)^p*(b+(c+(1-I*Sqrt[3])*r/2)*x)^p*(b+(c+(1+I*Sqrt[3])*r/2)*x)^p,x]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && NeQ[c^2-3*b*d,0] && EqQ[b^2-3*a*c,0]
```

2. $\int(e+f\,x)^m\,\left(a+b\,x+c\,x^2+d\,x^3\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,c^2-3\,b\,d\neq0\,\land\,b^2-3\,a\,c\neq0$

   1: $\int(e+f\,x)^m\,\left(a+b\,x+c\,x^2+d\,x^3\right)^p\,dx$ when $p\notin\mathbb{Z}\,\land\,c^2-3\,b\,d\neq0\,\land\,b^2-3\,a\,c\neq0\,\land\,\text{Factor}\left[a+b\,x+c\,x^2+d\,x^3\right]=v\,w\cdots$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $\text{Factor}\left[a+b\,x+c\,x^2+d\,x^3\right]=v\,w\cdots$, **then** $\partial_x\frac{\left(a+b\,x+c\,x^2+d\,x^3\right)^p}{v^p\,w^p\cdots}=0$
- **Rule: If** $p\notin\mathbb{Z}\,\land\,c^2-3\,b\,d\neq0\,\land\,b^2-3\,a\,c\neq0\,\land\,\text{Factor}\left[a+b\,x+c\,x^2+d\,x^3\right]=v\,w\cdots$, **then**

$$\int(e+f\,x)^m\,\left(a+b\,x+c\,x^2+d\,x^3\right)^p\,dx\;\longrightarrow\;\frac{\left(a+b\,x+c\,x^2+d\,x^3\right)^p}{v^p\,w^p\cdots}\int(e+f\,x)^m\,v^p\,w^p\cdots\,dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{u=NonfreeFactors[Factor[a+b*x+c*x^2+d*x^3],x]},
  (a+b*x+c*x^2+d*x^3)^p/DistributeDegree[u,p]*Int[(e+f*x)^m*DistributeDegree[u,p],x] /;
 ProductQ[u]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

x: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z}$?? ??

- **Derivation: Integration by substitution**
- **Basis: If** $p \in \mathbb{Z}$, **then** $\left(a+b x+c x^2+d x^3\right)^p = \frac{1}{3^{3p} d^{2p}} \text{Subst}\left[\left(2 c^3-9 b c d+27 a d^2-9 d \left(c^2-3 b d\right) x+27 d^3 x^3\right)^p, x, x+\frac{c}{3 d}\right] \partial_x \left(x+\frac{c}{3 d}\right)$
- **Note: The substitution transforms the cubic polynomial in the integrand into a reduced cubic of the form** $r+s x+t x^3$.
- **Rule: If** $p \notin \mathbb{Z}$, **then**

$$\int \left(a+b x+c x^2+d x^3\right)^p \, dx \longrightarrow \frac{1}{3^{3p} d^{2p}} \text{Subst}\left[\int \left(2 c^3-9 b c d+27 a d^2-9 d \left(c^2-3 b d\right) x+27 d^3 x^3\right)^p dx, x, x+\frac{c}{3 d}\right]$$

- **Program code:**

```
(* Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  1/(3^(3*p)*d^(2*p))*Subst[Int[(2*c^3-9*b*c*d+27*a*d^2-9*d*(c^2-3*b*d)*x+27*d^3*x^3)^p,x],x,x+c/(3*d)] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IntegerQ[p]] *)
```

2: $\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx$ when $p \notin \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$

- **Derivation: Piecewise constant extraction**
- **Basis: If** $r = \left(-2 c^3+9 b c d-27 a d^2+3 \sqrt{3}\, d \sqrt{-b^2 c^2+4 a c^3+4 b^3 d-18 a b c d+27 a^2 d^2}\right)^{1/3}$, **then**

$$\partial_x \left(a+b x+c x^2+d x^3\right)^p \Big/ \left(\left(c-\frac{2 c^2-6 b d+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(c+\frac{2 \left(1+i \sqrt{3}\right) c^2-6 \left(1+i \sqrt{3}\right) b d-i\, 2^{1/3} \left(i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2 \left(1-i \sqrt{3}\right) c^2-6 \left(1-i \sqrt{3}\right) b d+i\, 2^{1/3} \left(-i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p\right) = 0$$

- **Rule: If** $p \notin \mathbb{Z} \land c^2-3 b d \neq 0 \land b^2-3 a c \neq 0$, **let** $r = \left(-2 c^3+9 b c d-27 a d^2+3 \sqrt{3}\, d \sqrt{-b^2 c^2+4 a c^3+4 b^3 d-18 a b c d+27 a^2 d^2}\right)^{1/3}$, **then**

$$\int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p \, dx \longrightarrow$$

$$\frac{\left(a+b x+c x^2+d x^3\right)^p}{\left(c-\frac{2 c^2-6 b d+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p \left(c+\frac{2 \left(1+i \sqrt{3}\right) c^2-6 \left(1+i \sqrt{3}\right) b d-i\, 2^{1/3} \left(i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2 \left(1-i \sqrt{3}\right) c^2-6 \left(1-i \sqrt{3}\right) b d+i\, 2^{1/3} \left(-i+\sqrt{3}\right) r^2}{2 \times 2^{2/3} r}+3 d x\right)^p}\cdot$$

$$\int (e+f x)^m \left(c-\frac{2 c^2-6 b d+2^{1/3} r^2}{2^{2/3} r}+3 d x\right)^p$$
$$\left(c+\frac{2\left(1+i \sqrt{3}\right) c^2-6\left(1+i \sqrt{3}\right) b d-i\, 2^{1/3}\left(i+\sqrt{3}\right) r^2}{2\times 2^{2/3} r}+3 d x\right)^p \left(c+\frac{2\left(1-i \sqrt{3}\right) c^2-6\left(1-i \sqrt{3}\right) b d+i\, 2^{1/3}\left(-i+\sqrt{3}\right) r^2}{2\times 2^{2/3} r}+3 d x\right)^p dx$$

- **Program code:**

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*x_+c_.*x_^2+d_.*x_^3)^p_,x_Symbol] :=
  With[{r=Rt[-2*c^3+9*b*c*d-27*a*d^2+3*Sqrt[3]*d*Sqrt[-b^2*c^2+4*a*c^3+4*b^3*d-18*a*b*c*d+27*a^2*d^2],3]},
  (a+b*x+c*x^2+d*x^3)^p/
    ((c-(2*c^2-6*b*d+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2-6*(1+I*Sqrt[3])*b*d-I*2^(1/3)*(I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2-6*(1-I*Sqrt[3])*b*d+I*2^(1/3)*(-I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p)*
    Int[(e+f*x)^m*(c-(2*c^2-6*b*d+2^(1/3)*r^2)/(2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1+I*Sqrt[3])*c^2-6*(1+I*Sqrt[3])*b*d-I*2^(1/3)*(I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p*
      (c+(2*(1-I*Sqrt[3])*c^2-6*(1-I*Sqrt[3])*b*d+I*2^(1/3)*(-I+Sqrt[3])*r^2)/(2*2^(2/3)*r)+3*d*x)^p,x]] /;
FreeQ[{a,b,c,d,e,f,m,p},x] && Not[IntegerQ[p]] && NeQ[c^2-3*b*d,0] && NeQ[b^2-3*a*c,0]
```

3: $\int u^m v^p \, dx$ when $u == e + f x \wedge v == a + b x + c x^2 + d x^3$

- **Derivation: Algebraic normalization**
- **Rule: If** $u == e + f x \wedge v == a + b x + c x^2 + d x^3$**, then**

$$\int u^m v^p \, dx \longrightarrow \int (e+f x)^m \left(a+b x+c x^2+d x^3\right)^p dx$$

- **Program code:**

```
Int[u_^m_.*v_^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*ExpandToSum[v,x]^p,x] /;
FreeQ[{m,p},x] && LinearQ[u,x] && PolyQ[v,x,3] && Not[LinearMatchQ[u,x] && CubicMatchQ[v,x]]
```